These are the examples I showed in class on Tuesday October 14th.
Michael Monagan

```
> v := Vector([1,1]);
```

$$v := \begin{bmatrix} 1 \\ 1 \end{bmatrix}$$

```
> w := <2,1>;
```

$$w := \begin{bmatrix} 2 \\ 1 \end{bmatrix}$$

```
> v+w;
```

$$\begin{bmatrix} 3 \\ 2 \end{bmatrix}$$

```
> A := Matrix([[1,1],[1,0]]);
```

$$A := \begin{bmatrix} 1 & 1 \\ 1 & 0 \end{bmatrix}$$

```
> 2*A;
```

$$\begin{bmatrix} 2 & 2 \\ 2 & 0 \end{bmatrix}$$

```
> A.A;
```

$$\begin{bmatrix} 2 & 1 \\ 1 & 1 \end{bmatrix}$$

```
> A.v;
```

$$\begin{bmatrix} 2 \\ 1 \end{bmatrix}$$

```
> 1/A;
```

$$\begin{bmatrix} 0 & 1 \\ 1 & -1 \end{bmatrix}$$

```
> with(LinearAlgebra):
> Determinant(A);
```

$$-1$$

Two ways to solve $A \cdot x = b$ for $x \in \mathbb{R}^2$

```
> x := LinearSolve(A,w);
```

$$x := \begin{bmatrix} 1 \\ 1 \end{bmatrix}$$

```
> x := (1/A) . w;
```

$$x := \begin{bmatrix} 1 \\ 1 \end{bmatrix}$$

```
> A.x=w;
```

$$\begin{bmatrix} 2 \\ 1 \end{bmatrix} = \begin{bmatrix} 2 \\ 1 \end{bmatrix}$$

```
> with(GraphTheory):
> G := Graph( { {1,2},{2,3},{3,4},{4,1} } );
```

G := Graph 3: an undirected unweighted graph with 4 vertices and 4 edge(s)

```
> DrawGraph(G,style=circle);
```




```
> Edges(G);
```

$$\{\{1, 2\}, \{1, 4\}, \{2, 3\}, \{3, 4\}\}$$

```
> Vertices(G);
```

$$[1, 2, 3, 4]$$

```
> IsConnected(G);
```

true

```
> IsDirected(G);
```

false

```
> H := Graph( [P1,P2,P3,P4,P5], { [P1,P2], [P2,P1], [P2,P3], [P2,P4],
  [P3,P4], [P3,P5], [P5,P1], [P5,P4] } );
```

H := Graph 4: a directed unweighted graph with 5 vertices and 8 arc(s)

```
> DrawGraph(H,style=circle);
```




```
> IsDirected(H);
```

true

```
> Vertices(H);
```

[*P1*, *P2*, *P3*, *P4*, *P5*]

```
> Edges(H);
```

{[*P1*, *P2*], [*P2*, *P1*], [*P2*, *P3*], [*P2*, *P4*], [*P3*, *P4*], [*P3*, *P5*], [*P5*, *P1*], [*P5*, *P4*]}

```
> IsConnected(H);
```

true

```
> IsStronglyConnected(H);
```

false

There is no way to get from vertex P4 to any of the other vertices so the graph is not strongly connected. If we add the edge P4 -> P5 then we have a cycle P1 -> P2 -> P3 -> P4 -> P5 -> P1 so we can now get from any vertex to any other vertex

```
> AddArc(H,[P4,P5]);
```

Graph 4: a directed unweighted graph with 5 vertices and 9 arc(s)

```
> DrawGraph(H,style=circle);
```



```
> IsStronglyConnected(H);
```

$$true$$

```
> Arrivals(H);
```

$$[[P2, P5], [P1], [P2], [P2, P3, P5], [P3, P4]]$$

```
> Departures(H);
```

$$[[P2], [P1, P3, P4], [P4, P5], [P5], [P1, P4]]$$

```
> A := Matrix( [[0,1/3,0,0,1/2],
               [1,0,0,0,0],
               [0,1/3,0,0,0],
               [0,1/3,1/2,0,1/2],
               [0,0,1/2,1,0]] );
```

$$A := \begin{bmatrix} 0 & \frac{1}{3} & 0 & 0 & \frac{1}{2} \\ 1 & 0 & 0 & 0 & 0 \\ 0 & \frac{1}{3} & 0 & 0 & 0 \\ 0 & \frac{1}{3} & \frac{1}{2} & 0 & \frac{1}{2} \\ 0 & 0 & \frac{1}{2} & 1 & 0 \end{bmatrix}$$

```
> Q[0] := Vector([1/5,1/5,1/5,1/5,1/5]):
> Q[1] := A.Q[0];
```

$$Q_1 := \begin{bmatrix} \frac{1}{6} \\ \frac{1}{5} \\ \frac{1}{15} \\ \frac{4}{15} \\ \frac{3}{10} \end{bmatrix}$$

```
> for i to 10 do Q[i] := A.Q[i-1]; od:
> Q[10];
```

$$\begin{bmatrix} \frac{3257}{15552} \\ \frac{7937}{38880} \\ \frac{2723}{38880} \\ \frac{6307}{25920} \\ \frac{3539}{12960} \end{bmatrix}$$

```
> evalf(%,4);
```

$$\begin{bmatrix} 0.2094 \\ 0.2041 \\ 0.07004 \\ 0.2433 \\ 0.2731 \end{bmatrix}$$

```
> for i from 11 to 100 do Q[i] := A.Q[i-1]; od:
> evalf(Q[100],4);
```

$$\begin{bmatrix} 0.2069 \\ 0.2069 \\ 0.06897 \\ 0.2414 \\ 0.2759 \end{bmatrix}$$

```
> evalf(Q[99],4);
```

$$\begin{bmatrix} 0.2069 \\ 0.2069 \\ 0.06897 \\ 0.2414 \\ 0.2759 \end{bmatrix}$$